NAG Library Routine Document

F08GFF (DOPGTR)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08GFF (DOPGTR) generates the real orthogonal matrix Q, which was determined by F08GEF (DSPTRD) when reducing a symmetric matrix to tridiagonal form.

2 Specification

```
SUBROUTINE F08GFF(UPLO, N, AP, TAU, Q, LDQ, WORK, INFO)
INTEGER           N, LDQ, INFO
double precision  AP(*), TAU(*), Q(LDQ,*), WORK(N-1)
CHARACTER*1       UPLO
```

The routine may be called by its LAPACK name ***dopgtr***.

3 Description

F08GFF (DOPGTR) is intended to be used after a call to F08GEF (DSPTRD), which reduces a real symmetric matrix A to symmetric tridiagonal form T by an orthogonal similarity transformation: $A = QTQ^{\mathrm{T}}$. F08GEF (DSPTRD) represents the orthogonal matrix Q as a product of $n-1$ elementary reflectors.

This routine may be used to generate Q explicitly as a square matrix.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: this **must** be the same parameter UPLO as supplied to F08GEF (DSPTRD).

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix Q.

Constraint: $\mathrm{N} \geq 0$.

3: AP($*$) – ***double precision*** array *Input*

Note: the dimension of the array AP must be at least $\max(1, \mathrm{N} \times (\mathrm{N}+1)/2)$.

On entry: details of the vectors which define the elementary reflectors, as returned by F08GEF (DSPTRD).

4: TAU($*$) – ***double precision*** array *Input*

Note: the dimension of the array TAU must be at least $\max(1, \mathrm{N}-1)$.

On entry: further details of the elementary reflectors, as returned by F08GEF (DSPTRD).

5: Q(LDQ,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array Q must be at least $\max(1, \mathrm{N})$.

On exit: the n by n orthogonal matrix Q.

6: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08GFF (DOPGTR) is called.

Constraint: LDQ $\geq \max(1, \mathrm{N})$.

7: WORK(N $-$ 1) – ***double precision*** array *Workspace*

8: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed matrix Q differs from an exactly orthogonal matrix by a matrix E such that

$$\|E\|_2 = O(\epsilon),$$

where ϵ is the ***machine precision***.

8 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^3$.

The complex analogue of this routine is F08GTF (ZUPGTR).

9 Example

This example computes all the eigenvalues and eigenvectors of the matrix A, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage. Here A is symmetric and must first be reduced to tridiagonal form by F08GEF (DSPTRD). The program then calls F08GFF (DOPGTR) to form Q, and passes this matrix to F08JEF (DSTEQR) which computes the eigenvalues and eigenvectors of A.

9.1 Program Text

```
*     F08GFF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDQ
      PARAMETER        (NMAX=8,LDQ=NMAX)
```

```
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      DOUBLE PRECISION AP(NMAX*(NMAX+1)/2), D(NMAX), E(NMAX),
     +                 Q(LDQ,NMAX), TAU(NMAX), WORK(2*NMAX-2)
*     .. External Subroutines ..
      EXTERNAL         DOPGTR, DSPTRD, DSTEQR, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08GFF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Reduce A to tridiagonal form T = (Q**T)*A*Q
*
         CALL DSPTRD(UPLO,N,AP,D,E,TAU,INFO)
*
*        Form Q explicitly, storing the result in Q
*
         CALL DOPGTR(UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
*
*        Calculate all the eigenvalues and eigenvectors of A
*
         CALL DSTEQR('V',N,D,E,Q,LDQ,WORK,INFO)
*
         WRITE (NOUT,*)
         IF (INFO.GT.0) THEN
            WRITE (NOUT,*) 'Failure to converge.'
         ELSE
*
*           Print eigenvalues and eigenvectors
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (D(I),I=1,N)
            WRITE (NOUT,*)
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,N,Q,LDQ,'Eigenvectors',IFAIL)
*
         END IF
      END IF
*
99999 FORMAT (3X,(8F8.4))
      END
```

9.2 Program Data

```
F08GFF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F08GFF Example Program Results

 Eigenvalues
    -5.0034 -1.9987  0.2013  8.0008

 Eigenvectors
          1       2       3       4
 1   0.5658 -0.2328 -0.3965  0.6845
 2  -0.3478  0.7994 -0.1780  0.4564
 3  -0.4740 -0.4087  0.5381  0.5645
 4   0.5781  0.3737  0.7221  0.0676
```